

# NAG Library Routine Document

## D06ABF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

D06ABF generates a triangular mesh of a closed polygonal region in  $\mathbb{R}^2$ , given a mesh of its boundary. It uses a Delaunay–Voronoi process, based on an incremental method.

### 2 Specification

```

SUBROUTINE D06ABF (NVB, NVINT, NVMAX, NEDGE, EDGE, NV, NELT, COOR, CONN, &
                  WEIGHT, NPROPA, ITRACE, RWORK, LRWORK, IWORK, LIWORK, &
                  IFAIL)
INTEGER          NVB, NVINT, NVMAX, NEDGE, EDGE(3,NEDGE), NV, NELT, &
                  CONN(3,2*NVMAX+5), NPROPA, ITRACE, LRWORK, &
                  IWORK(LIWORK), LIWORK, IFAIL
REAL (KIND=nag_wp) COOR(2,NVMAX), WEIGHT(*), RWORK(LRWORK)

```

### 3 Description

D06ABF generates the set of interior vertices using a Delaunay–Voronoi process, based on an incremental method. It allows you to specify a number of fixed interior mesh vertices together with weights which allow concentration of the mesh in their neighbourhood. For more details about the triangulation method, consult the D06 Chapter Introduction as well as George and Borouchaki (1998).

This routine is derived from material in the MODULEF package from INRIA (Institut National de Recherche en Informatique et Automatique).

### 4 References

George P L and Borouchaki H (1998) *Delaunay Triangulation and Meshing: Application to Finite Elements* Editions HERMES, Paris

### 5 Parameters

- 1: NVB – INTEGER *Input*  
*On entry:* the number of vertices in the input boundary mesh.  
*Constraint:* NVB  $\geq$  3.
- 2: NVINT – INTEGER *Input*  
*On entry:* the number of fixed interior mesh vertices to which a weight will be applied.  
*Constraint:* NVINT  $\geq$  0.
- 3: NVMAX – INTEGER *Input*  
*On entry:* the maximum number of vertices in the mesh to be generated.  
*Constraint:* NVMAX  $\geq$  NVB + NVINT.

- 4: NEDGE – INTEGER *Input*  
*On entry:* the number of boundary edges in the input mesh.  
*Constraint:*  $NEDGE \geq 1$ .
- 5: EDGE(3, NEDGE) – INTEGER array *Input*  
*On entry:* the specification of the boundary edges. EDGE(1,  $j$ ) and EDGE(2,  $j$ ) contain the vertex numbers of the two end points of the  $j$ th boundary edge. EDGE(3,  $j$ ) is a user-supplied tag for the  $j$ th boundary edge and is not used by D06ABF.  
*Constraint:*  $1 \leq \text{EDGE}(i, j) \leq \text{NVB}$  and  $\text{EDGE}(1, j) \neq \text{EDGE}(2, j)$ , for  $i = 1, 2$  and  $j = 1, 2, \dots, \text{NEDGE}$ .
- 6: NV – INTEGER *Output*  
*On exit:* the total number of vertices in the output mesh (including both boundary and interior vertices). If  $\text{NVB} + \text{NVINT} = \text{NVMAX}$ , no interior vertices will be generated and  $\text{NV} = \text{NVMAX}$ .
- 7: NELT – INTEGER *Output*  
*On exit:* the number of triangular elements in the mesh.
- 8: COOR(2, NVMAX) – REAL (KIND=nag\_wp) array *Input/Output*  
*On entry:* COOR(1,  $i$ ) contains the  $x$  coordinate of the  $i$ th input boundary mesh vertex, for  $i = 1, 2, \dots, \text{NVB}$ . COOR(1,  $i$ ) contains the  $x$  coordinate of the  $(i - \text{NVB})$ th fixed interior vertex, for  $i = \text{NVB} + 1, \dots, \text{NVB} + \text{NVINT}$ . For boundary and interior vertices, COOR(2,  $i$ ) contains the corresponding  $y$  coordinate, for  $i = 1, 2, \dots, \text{NVB} + \text{NVINT}$ .  
*On exit:* COOR(1,  $i$ ) will contain the  $x$  coordinate of the  $(i - \text{NVB} - \text{NVINT})$ th generated interior mesh vertex, for  $i = \text{NVB} + \text{NVINT} + 1, \dots, \text{NV}$ ; while COOR(2,  $i$ ) will contain the corresponding  $y$  coordinate. The remaining elements are unchanged.
- 9: CONN(3,  $2 \times \text{NVMAX} + 5$ ) – INTEGER array *Output*  
*On exit:* the connectivity of the mesh between triangles and vertices. For each triangle  $j$ , CONN( $i, j$ ) gives the indices of its three vertices (in anticlockwise order), for  $i = 1, 2, 3$  and  $j = 1, 2, \dots, \text{NELT}$ .
- 10: WEIGHT(\*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the dimension of the array WEIGHT must be at least  $\max(1, \text{NVINT})$ .  
*On entry:* the weight of fixed interior vertices. It is the diameter of triangles (length of the longer edge) created around each of the given interior vertices.  
*Constraint:* if  $\text{NVINT} > 0$ ,  $\text{WEIGHT}(i) > 0.0$ , for  $i = 1, 2, \dots, \text{NVINT}$ .
- 11: NPROPA – INTEGER *Input*  
*On entry:* the propagation type and coefficient, the parameter NPROPA is used when the internal points are created. They are distributed in a geometric manner if NPROPA is positive and in an arithmetic manner if it is negative. For more details see Section 9.  
*Constraint:*  $\text{NPROPA} \neq 0$ .
- 12: ITRACE – INTEGER *Input*  
*On entry:* the level of trace information required from D06ABF.  
 $\text{ITRACE} \leq 0$   
 No output is generated.

ITRACE  $\geq 1$

Output from the meshing solver is printed on the current advisory message unit (see X04ABF). This output contains details of the vertices and triangles generated by the process.

You are advised to set ITRACE = 0, unless you are experienced with finite element mesh generation.

13: RWORK(LRWORK) – REAL (KIND=nag\_wp) array Workspace  
 14: LRWORK – INTEGER Input

*On entry:* the dimension of the array RWORK as declared in the (sub)program from which D06ABF is called.

*Constraint:* LRWORK  $\geq 12 \times NVMAX + 15$ .

15: IWORK(LIWORK) – INTEGER array Workspace  
 16: LIWORK – INTEGER Input

*On entry:* the dimension of the array IWORK as declared in the (sub)program from which D06ABF is called.

*Constraint:* LIWORK  $\geq 6 \times NEDGE + 32 \times NVMAX + 2 \times NVB + 78$ .

17: IFAIL – INTEGER Input/Output

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NVB < 3,  
 or NVINT < 0,  
 or NVB + NVINT > NVMAX,  
 or NEDGE < 1,  
 or EDGE(*i*, *j*) < 1 or EDGE(*i*, *j*) > NVB, for some *i* = 1, 2 and *j* = 1, 2, ..., NEDGE,  
 or EDGE(1, *j*) = EDGE(2, *j*), for some *j* = 1, 2, ..., NEDGE,  
 or NPROPA = 0;  
 or if NVINT > 0, WEIGHT(*i*)  $\leq 0.0$ , for some *i* = 1, 2, ..., NVINT;  
 or LRWORK < 12  $\times$  NVMAX + 15,  
 or LIWORK < 6  $\times$  NEDGE + 32  $\times$  NVMAX + 2  $\times$  NVB + 78.

IFAIL = 2

An error has occurred during the generation of the interior mesh. Check the definition of the boundary (arguments COOR and EDGE) as well as the orientation of the boundary (especially in

the case of a multiple connected component boundary). Setting ITRACE > 0 may provide more details.

IFAIL = 3

An error has occurred during the generation of the boundary mesh. It appears that NVMAX is not large enough.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.  
See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.  
See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.  
See Section 3.6 in the Essential Introduction for further information.

## 7 Accuracy

Not applicable.

## 8 Parallelism and Performance

D06ABF is not threaded by NAG in any implementation.

D06ABF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The position of the internal vertices is a function position of the vertices on the given boundary. A fine mesh on the boundary results in a fine mesh in the interior. To dilute the influence of the data on the interior of the domain, the value of NPROPA can be changed. The propagation coefficient is calculated as:  $\omega = 1 + \frac{a - 1.0}{20.0}$ , where  $a$  is the absolute value of NPROPA. During the process vertices are generated on edges of the mesh  $\mathcal{T}_i$  to obtain the mesh  $\mathcal{T}_{i+1}$  in the general incremental method (consult the D06 Chapter Introduction or George and Borouchaki (1998)). This generation uses the coefficient  $\omega$ , and it is geometric if NPROPA > 0, and arithmetic otherwise. But increasing the value of  $a$  may lead to failure of the process, due to precision, especially in geometries with holes. So you are advised to manipulate the parameter NPROPA with care.

You are advised to take care to set the boundary inputs properly, especially for a boundary with multiply connected components. The orientation of the interior boundaries should be in **clockwise** order and opposite to that of the exterior boundary. If the boundary has only one connected component, its orientation should be **anticlockwise**.

## 10 Example

In this example, a geometry with two holes (two wings inside an exterior circle) is meshed using a Delaunay–Voronoi method. The exterior circle is centred at the point (1.0,0.0) with a radius 3. The main wing, using aerofoil RAE 2822 data, lies between the origin and the centre of the circle, while the secondary aerofoil is produced from the first by performing a translation, a scale reduction and a rotation. To be able to carry out some realistic computation on that geometry, some interior points have been introduced to have a finer mesh in the wake of those aerofoils.

The boundary mesh has 296 vertices and 296 edges (see Section 10.3 top). Note that the particular mesh generated could be sensitive to the *machine precision* and therefore may differ from one implementation to another. The interior meshes for different values of NPROPA are given in Section 10.3.

### 10.1 Program Text

```

Program d06abfe

!      D06ABF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: d06abf, f06epf, nag_wp, x01aaf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: meshout = 7, nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: c, d, dnvint, r, s, theta, theta_i
Integer                    :: i, il, ic, ifail, itrace, j, liwork, &
                             lrwork, nearest, nedge, nelt,      &
                             nelt_near, npropa, nrae, nv, nvb,   &
                             nvint, nvmax, nv_near
Character (1)              :: pmesh
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: coor(:,,:), rwork(:), weight(:)
Real (Kind=nag_wp)             :: t(2)
Integer, Allocatable           :: conn(:,,:), edge(:,,:), iwork(:)
!      .. Intrinsic Procedures ..
Intrinsic                     :: cos, real, sin, tan
!      .. Executable Statements ..
Write (nout,*) 'D06ABF Example Program Results'

!      Skip heading in data file
Read (nin,*)

!      Reading of the geometry:
!      number of points on RAE aerofoil data;
!      number of points on circular boundary;
!      maximum number of vertices.

Read (nin,*) nrae, nvint, nvmax
Read (nin,*) pmesh

nvb = nvint + 2*nrae
nedge = nvb

lrwork = 12*nvmax + 15
liwork = 6*nedge + 32*nvmax + 2*nvb + 78
Allocate (coor(2,nvmax),rwork(lrwork),weight(nvint),conn(3,2*nvmax+5), &
          edge(3,nedge),iwork(liwork))

!      Circular outer boundary, radius 3 and centre (1,0)
theta = 2.0_nag_wp*x01aaf(r)/real(nvint,kind=nag_wp)
r = 3.0_nag_wp
t(1) = 1.0_nag_wp
Do i = 1, nvint
    theta_i = theta*real(i-1,kind=nag_wp)

```

```

        coor(1,i) = r*cos(theta_i) + t(1)
        coor(2,i) = r*sin(theta_i)
    End Do

!   Read data for aerofoil RAE 2822
    Do i = 1, nrae
        Read (nin,*) il, coor(1,nvint+i), coor(2,nvint+i)
    End Do

!   Transform RAE 2822 for secondary foil
    theta = x01aaf(theta)/12.0_nag_wp
    c = cos(theta)
    s = sin(theta)
    ic = nvint + nrae
!   Copy and rotate coordinates by theta = pi/12
    coor(1:2,ic+1:ic+nrae) = coor(1:2,nvint+1:ic)
    Call f06epf(nrae,coor(1,ic+1),2,coor(2,ic+1),2,c,s)
!   Reduce by 0.4 and translate to distance 0.25 from intercept at (0.75,0)
    d = 0.4_nag_wp
    t(1) = 0.75_nag_wp + 0.25_nag_wp*c
    t(2) = -0.25_nag_wp*s
    Do i = 1, nrae
        coor(1:2,ic+i) = d*coor(1:2,ic+i) + t(1:2)
    End Do

!   Boundary edges
    Do i = 1, nedge
        edge(1,i) = i
        edge(2,i) = i + 1
        edge(3,i) = 0
    End Do
!   Tie up end of three boundary edges
    edge(2,nvint) = 1
    edge(2,nvint+nrae) = nvint + 1
    edge(2,nedge) = nvint + nrae + 1

!   Initialise mesh control parameters

    itrace = 0

!   Generation of interior vertices on the
!   RAE airfoil's wake

    dnvint = 2.5E0_nag_wp/real(nvint+1,kind=nag_wp)

    Do i = 1, nvint
        il = nvb + i
        coor(1,il) = 1.38E0_nag_wp + real(i,kind=nag_wp)*dnvint
        coor(2,il) = -tan(theta)*(coor(1,il)-0.75_nag_wp)
    End Do

    weight(1:nvint) = 0.01E0_nag_wp

    Write (nout,*)

!   Loop on the propagation coef
pcoef: Do j = 1, 4

        nearest = 200
        Select Case (j)
        Case (1)
            npropa = -5
            nearest = 150
        Case (2)
            npropa = -1
        Case (3)
            npropa = 1
        Case Default
            npropa = 5
        End Select

```

```

!      Call to the 2D Delaunay-Voronoi mesh generator

      ifail = 0
      Call d06abf(nvb,nvint,nvmax,nedge,edge,nv,nelt,coor,conn,weight, &
        npropa,itrace,rwork,lrwork,iwork,liwork,ifail)

      Write (nout,99999) 'Mesh characteristics with NPROPA =', npropa
      nv_near = ((nv+nearest/2)/nearest)*nearest
      nelt_near = ((nelt+nearest/2)/nearest)*nearest
      Write (nout,99998) 'NV ', nv_near, nearest
      Write (nout,99998) 'NELT', nelt_near, nearest

      If (pmesh=='Y') Then
!      Output the mesh in a form suitable for printing

        If (j==1) Then
          Write (meshout,*) '# D06ABF Example Program Mesh results'
        End If
        Write (meshout,99999) '# Mesh line segments for NPROPA =', npropa
        Do i = 1, nelt
          Write (meshout,99997) coor(1,conn(1,i)), coor(2,conn(1,i))
          Write (meshout,99997) coor(1,conn(2,i)), coor(2,conn(2,i))
          Write (meshout,99997) coor(1,conn(3,i)), coor(2,conn(3,i))
          Write (meshout,99997) coor(1,conn(1,i)), coor(2,conn(1,i))
          Write (meshout,*)
        End Do
        Write (meshout,*)
      End If

      End Do pcoef

99999 Format (1X,A,I6)
99998 Format (1X,A5,' = ',I10,' to the nearest ',I3)
99997 Format (2(2X,E13.6))
      End Program d06abfe

```

## 10.2 Program Data

**Note 1:** since the data file for this example is quite large only a section of it is reproduced in this document. The full data file is distributed with your implementation.

```

D06ABF Example Program Data
128      40      6000      : nrae nvint nvmax
'N'      : Printing mesh? 'Y' or 'N'

01      0.000000E+00 0.000000E+00
02      0.602000E-03 0.316500E-02
03      0.240800E-02 0.630600E-02
04      0.541200E-02 0.941600E-02
05      0.960700E-02 0.124800E-01
06      0.149840E-01 0.154890E-01
07      0.215300E-01 0.184410E-01
08      0.292280E-01 0.213480E-01
09      0.380600E-01 0.242190E-01
10      0.480050E-01 0.270620E-01
11      0.590390E-01 0.298740E-01
12      0.711360E-01 0.326440E-01
13      0.842650E-01 0.353600E-01
14      0.983960E-01 0.380110E-01
15      0.113495E+00 0.405850E-01
16      0.129524E+00 0.430710E-01
17      0.146447E+00 0.454570E-01
18      0.164221E+00 0.477290E-01
19      0.182803E+00 0.498740E-01
20      0.202150E+00 0.518850E-01
21      0.222215E+00 0.537530E-01
22      0.242949E+00 0.554700E-01
23      0.264302E+00 0.570260E-01
24      0.286222E+00 0.584140E-01

```

25	0.308658E+00	0.596290E-01
26	0.331555E+00	0.606600E-01
27	0.354858E+00	0.614970E-01
28	0.378510E+00	0.621330E-01
29	0.402455E+00	0.625620E-01
30	0.426635E+00	0.627790E-01
31	0.450991E+00	0.627740E-01
32	0.475466E+00	0.625300E-01
33	0.500000E+00	0.620290E-01
34	0.524534E+00	0.612540E-01
35	0.549009E+00	0.601940E-01
36	0.573365E+00	0.588450E-01
37	0.597545E+00	0.572180E-01
38	0.621490E+00	0.553440E-01
39	0.645142E+00	0.532580E-01
40	0.668445E+00	0.509930E-01
41	0.691342E+00	0.485750E-01
42	0.713778E+00	0.460290E-01
43	0.735698E+00	0.433770E-01
44	0.757051E+00	0.406410E-01
45	0.777785E+00	0.378470E-01
46	0.797850E+00	0.350170E-01
47	0.817197E+00	0.321760E-01
48	0.835779E+00	0.293470E-01
49	0.853553E+00	0.265540E-01
50	0.870476E+00	0.238170E-01
51	0.886505E+00	0.211530E-01
52	0.901604E+00	0.185800E-01
53	0.915735E+00	0.161130E-01
54	0.928864E+00	0.137690E-01
55	0.940961E+00	0.115620E-01
56	0.951995E+00	0.950800E-02
57	0.961940E+00	0.762200E-02
58	0.970772E+00	0.591500E-02
59	0.978470E+00	0.440100E-02
60	0.985016E+00	0.309200E-02
61	0.990393E+00	0.200100E-02
62	0.994588E+00	0.113700E-02
63	0.997592E+00	0.510000E-03
64	0.999398E+00	0.128000E-03
65	0.100000E+01	0.000000E+00
66	0.999398E+00	0.350000E-04
67	0.997592E+00	0.137000E-03
68	0.994588E+00	0.296000E-03
69	0.990393E+00	0.497000E-03
70	0.985016E+00	0.719000E-03
71	0.978470E+00	0.935000E-03
72	0.970772E+00	0.111200E-02
73	0.961940E+00	0.121200E-02
74	0.951995E+00	0.119700E-02
75	0.940961E+00	0.103300E-02
76	0.928864E+00	0.694000E-03
77	0.915735E+00	0.157000E-03
78	0.901604E+00	-.600000E-03
79	0.886505E+00	-.159200E-02
80	0.870476E+00	-.282900E-02
81	0.853553E+00	-.431400E-02
82	0.835779E+00	-.604800E-02
83	0.817197E+00	-.802700E-02
84	0.797850E+00	-.102440E-01
85	0.777785E+00	-.126900E-01
86	0.757051E+00	-.153570E-01
87	0.735698E+00	-.182320E-01
88	0.713778E+00	-.212890E-01
89	0.691342E+00	-.244950E-01
90	0.668445E+00	-.278140E-01
91	0.645142E+00	-.312070E-01
92	0.621490E+00	-.346310E-01
93	0.597545E+00	-.380430E-01
94	0.573365E+00	-.413970E-01
95	0.549009E+00	-.446420E-01



96	0.524534E+00	-.477190E-01	
97	0.500000E+00	-.505630E-01	
98	0.475466E+00	-.530990E-01	
99	0.450991E+00	-.552570E-01	
100	0.426635E+00	-.569790E-01	
101	0.402455E+00	-.582240E-01	
102	0.378510E+00	-.589740E-01	
103	0.354858E+00	-.592360E-01	
104	0.331555E+00	-.590460E-01	
105	0.308658E+00	-.584590E-01	
106	0.286222E+00	-.575470E-01	
107	0.264302E+00	-.563760E-01	
108	0.242949E+00	-.549940E-01	
109	0.222215E+00	-.534270E-01	
110	0.202150E+00	-.516940E-01	
111	0.182803E+00	-.498050E-01	
112	0.164221E+00	-.477730E-01	
113	0.146447E+00	-.456100E-01	
114	0.129524E+00	-.433260E-01	
115	0.113495E+00	-.409290E-01	
116	0.983960E-01	-.384310E-01	
117	0.842650E-01	-.358430E-01	
118	0.711360E-01	-.331700E-01	
119	0.590390E-01	-.304160E-01	
120	0.480050E-01	-.275860E-01	
121	0.380600E-01	-.246850E-01	
122	0.292280E-01	-.217220E-01	
123	0.215300E-01	-.187070E-01	
124	0.149840E-01	-.156490E-01	
125	0.960700E-02	-.125590E-01	
126	0.541200E-02	-.944300E-02	
127	0.240800E-02	-.630800E-02	
128	0.602000E-03	-.316000E-02	: RAE Aerofoil 2822

### 10.3 Program Results


D06ABF Example Program Results

```


Mesh characteristics with NPROPA = -5
  NV = 2250 to the nearest 150
  NELT = 4350 to the nearest 150
Mesh characteristics with NPROPA = -1
  NV = 4400 to the nearest 200
  NELT = 8600 to the nearest 200
Mesh characteristics with NPROPA = 1
  NV = 5200 to the nearest 200
  NELT = 10000 to the nearest 200
Mesh characteristics with NPROPA = 5
  NV = 2000 to the nearest 200
  NELT = 3800 to the nearest 200

```


**Example Program**  
Geometry for Generating Meshes



Mesh Generated Using Arithmetic Coefficient  $\omega=1.2$



Mesh Generated Using Arithmetic Coefficient  $\omega=1.0$



Mesh Generated Using Geometric Coefficient  $\omega=1.0$

